
Széchenyi István Egyetem  Alkalmazott Mechanika  

 Műszaki Tudományi Kar  Tanszék 

 GÉPEK DINAMIKÁJA     4.gyak.hét – 1. és 2. Feladat 

(kidolgozta: Dr. Nagy Zoltán egyetemi adjunktus) 

4. gyak. hét – 1. feladat (Runge-Kutta módszer) 

Program begépelése: 

// 4.hét - 1.feladat 

clear all; 

usecanvas(%T); 

// Másodrendű állandó együtthatós differenciálegyenlet megoldása 

// Runge-Kutta módszerrel :  Jaz*(y")=-lSA*mr*g*sin(y) . 

// Átrendezés:  y"=-(lSA*mr*g/Jaz)*sin(y)  ....  ezt kell megoldanunk. 

// A megoldás kezdeti feltételei:  

//  t=0 (s)  y(0)=0.3 (rad) kezdeti szögelfordulás ,  

//  t=0 (s)  y'(0)=0.5 (rad/s) kezdeti szögsebesség. 

//------------------------------------------------------------------------------------- 

// legyen a megoldás két elemű oszlopvektora: p , ahol 

// p(1)=y (rad) szögelfordulás, 

//  p(2)=y' (rad/s) szögsebesség. 

// ennek derivált vektora: p' , ahol 

//  p'(1)=y' azaz p(2) (rad/s) szögsebesség, 

// p'(2)=y" (rad/s^2) szöggyorsulás, 

//  p'(2)=y"=-(lSA*mr*g/Jaz)*sin(y) ,  

//  p'(2)=y"=-(lSA*mr*g/Jaz)*sin(p1).  

//------------------BEMENŐ ADATOK ÉRTÉKEI ----------------------------- 

lSA=0.15;  // (m) 

mr=0.45;  // (kg) 

JAZ=0.05;  // (kgm^2) 

g=10.0;  // (m/s^2) 

//------------------------------------------------------------------------------------- 

// 

function [pdot]=f(t, p) 

// a p' , azaz a derivált vektor megadása következik,  

// amit majd integrálunk !!!! 

pdot(1)=p(2)  // sebesség, ezt mindig így, ilyen formában kell megadni !! 

pdot(2)=-(lSA*mr*g/JAZ)*sin(p(1))  // a fgv. gyorsulásra kifejezett alakja 

endfunction 

// 

// Diff. egyenlet kezdeti feltételeinek megadása az előírások alapján 

t0=0;  // kezdeti időpont zérus 

p0(1)=0.3  // kezdeti szögelfordulás radiánban ( = 17.191 fok) 

p0(2)=0.5;  // kezdeti szögsebesség, azaz y'(0)= 0.5 (rad/s). 

// Megoldás időintervalluma 

t=0:0.001:10;  // 10 másodperc, 0,001 -es lépésközzel végrehajtva 

// 

// Runge-Kutta módszer aktivizálása 

p=ode("rk",p0,t0,t,f); 

//////////////////////////  VIGYÁZAT !!!  ////////////////////// 

// Ezzel a módszerrel a szöggyorsulás függvényt nem tudjuk kiíratni, hiszen  

// a megoldást jelentő p vektor első eleme az szögelfordulás , a második eleme 

// pedig a szögsebesség értékeit tartalmazza! Amennyiben a szöggyorsulást is  

// szeretnénk kiíratni, így ezt újból fel kell írni a programban! 

// 

a=-(lSA*mr*g/JAZ)*sin(p(1,:));  // szöggyorsulás fgv (rad/s^2)  

// 

// Eredmények rajzoltatása 

subplot(3,1,1) 

plot2d(t,p(1,:)*(180/(%pi)),1);  // a szögelfordulást fokban ábrázoljuk 

xtitle(" szögelfordulás fgv","t(s)","kitérés (fok)"); 

xgrid(2); 

// 

subplot(3,1,2) 

plot2d(t,p(2,:),5); 

xtitle("szögsebesség fgv","t(s)","szögsebesség (rad/s)"); 

xgrid(2); 

// 

subplot(3,1,3) 

plot2d(t,a(1,:),1); 

xtitle("szöggyorsulás fgv","t(s)","szöggyorsulás (rad/s^2)"); 

xgrid(2); 

Program futtatása: 

 

 

4.gyak.hét –  2. feladat (Euler módszer) 

 

Program begépelése: 

// 4.hét - 2.feladat 

clear all; 

usecanvas(%T); 

stacksize('max'); 

// Másodrendű állandó együtthatós differenciálegyenlet megoldása 

// Euler módszerrel : JAz*(y")= -lSA*mr*g*sin(y)  

// Átrendezés:  y"=-(lSA*mr*g/JAz)*sin(y).... ezt kell megoldanunk. 

//------------------BEMENŐ ADATOK ÉRTÉKEI ---------------------------------------- 

lSA=0.15;  // (m) 

mr=0.45;  // (kg) 

JAz=0.05;  // (kgm^2) 

g=10.0;  // (m/s^2) 

// ------------- Megoldás időintervalluma ------------------------------------------------- 

dt=0.001; 

tmax=10.0; 

n=int(tmax/dt); 

//------------------------------------------------------------------------------------------------ 

// Diff. egyenlet kezdeti feltételeinek megadása ( új érték   kell)!!!! 

t0=0;  // kezdeti időpont zérus (s). 

y0=0.3;  // kezdeti szögelfordulás (rad). 

v0=0.5;    // kezdeti szögsebesség, azaz y'(0)= ..... (rad/s). 

a0=-(lSA*mr*g/JAz)*sin(y0);  // kezdeti szöggyorsulás: ezt számítani kell 

// ----------------------------------------------------------------------------------------------- 

t0=dt; 

t=(1:n); 

y=(1:n); 

v=(1:n); 

a=(1:n); 

// indítás pillanatában   //////  i=1  ciklus változó //////////// 

t(1)=0; 

y(1)=y0; 

v(1)=v0; 

a(1)=a0; 

/////////// a for ciklus kezdete //////////////////////////////// 

for i=2:n 

    t(i)=t0+dt; 

    a(i)=-(lSA*mr*g/JAz)*sin(y0);  

    v(i)=v0+((a(i)+a0)/2)*dt; 

    y(i)=y0+((v(i)+v0)/2)*dt; 

/// A következő, új ciklus kezdő értékei (értékcsere)                            

    t0=t(i); 

    y0=y(i); 

    v0=v(i); 

    a0=a(i); 

 end 

// 

// Eredmények rajzoltatása 

subplot(3,1,1) 

plot2d(t,y(1,:)*(180/(%pi)),1);  // szögelfordulás fokban ábrázolva 

xtitle(" szögelfordulás fgv","t(s)","kitérés (fok)"); 

xgrid(2); 

// 

subplot(3,1,2) 

plot2d(t,v(1,:),5); 

xtitle("szögsebesség fgv","t(s)","szögsebesség (rad/s)"); 

xgrid(2); 

// 

subplot(3,1,3) 

plot2d(t,a(1,:),1); 

xtitle("szöggyorsulás fgv","t(s)","szöggyorsulás (rad/s^2)"); 

xgrid(2); 
 
 

A program futtatása: 

 

 
 


